

SAMPLE PROGRAMS FOR NUMERICAL INTEGRATION:

Left and Right Hand Riemann Sums, Trapezoid Rule, Midpoint Rule, and Simpson's Rule. (L,R,T,M,S)

These programs are easily adaptable to other programmable calculators.

TI-81 Calculator

To use, simply enter $Y_1 = f(x)$. To start the program, press "PRGM", then the appropriate number for the INTEGRAL program, then "ENTER". At the prompts, enter the lower limit, the upper limit, and the number of divisions. Pressing "ENTER" again will allow you to input a different value for the number of divisions. To exit, press "ON", then "2". Pressing "ENTER" again will restart the program to allow you to re-enter new values.

```
Prgm1:INTEGRAL
:Disp "LOWER"
:Input A
:Disp "UPPER"
:Input B
:Lbl 2
:Disp "N"
:Input N
:(B-A)/2N→H
:0→L
:0→M
:1→J
:A→X
:Lbl 1
:L+Y1→L
:X+H→X
:M+Y1→M
:X+H→X
:IS>(J,N)
:Goto 1
:Disp"L/R/T"
:2HL→L
:Disp L
:L+2HY1→T
:A→X
:T-2HY1→T
:Disp T
:(L+T)/2→T
:Disp T
:Disp "MID/SIM"
:2HM→M
:Disp M
:(2M+T)/3→S
:Disp S
:Pause
:Goto 2
:
```

TI-85 Calculator

To use, enter $y_1 = f(x)$. To start the program, press "PRGM", then "NAMES", then "INTEG". At the prompts, enter the lower limit, the upper limit, and the number of divisions. Pressing "ENTER" again will restart the program to allow you to re-enter new values.

```
Prgm:INTEGRAL
:Disp "LOWER LIMIT"
:Input A
:Disp "UPPER LIMIT"
:Input B
:Disp "DIVISIONS"
:Input N
:(B-A)/N→H
:A→x
:0→L
:0→M
:1→I
:Lbl P
:L+H*y1→L
:x+0.5H→x
:M+H*y1→M
:x+0.5H→x
:IS>(I,N)
:Goto P
:Disp"LEFT/RIGHT"
:Disp L
:L+H*y1→R
:A→x
:R-H*y1→R
:Disp R
:(L+R)/2→T
:Disp "TRAP/MID/SIMP"
:Disp T
:Disp M
:(2M+T)/3→S
:Disp S
:
```